
Bevezetés

Három térdimenzióig a Galilei-csoport gyenge kohomológiájának jellem-
zése már ismert. Jelen szakdolgozattal magasabb térdimenziós esetekben is
meghatározzuk a Galilei-csoport gyenge kohomológiáinak jellemzésére alkal-
mas paramétereket, sőt egy Maple 18 programkódot is közlünk amit arra
írtunk, hogy segítségével meghatározhassuk tetszőleges Galilei-csoport gyen-
ge kohomológiájának jellemzésére alkalmas paramétereket.

Az 1. fejezetben bevezetjük a Galilei-csoport definícióját, és megismerke-
dünk alapvető tulajdonságaival. Definiáljuk a témakör tárgyalásához szük-
séges alapfogalmakat, valamint meghatározzuk a Lie-algebrát. Továbbá be-
vezetünk a Galilei-csoport megértéséhez szükséges fogalmakat. Megismerke-
dünk a félcsoportokkal, az egyparaméteres részcsoportokkal, a kommutátor
kociklusokkal, s majd a speciális lineáris csoporton keresztül szemléltetjük
őket.

A 2. fejezetben a legfeljebb három térdimenziós Galilei-csoportok kommu-
tátor kociklusainak gyenge kohomológiáit meghatározó paramétereket szá-
mítjuk ki. A 3. fejezetben pedig bemutatjuk az imént már említett számí-
tógépes programot, aminek segítségével 10 tér dimenzióig meghatározzuk a
Galilei-csoport gyenge kohomológiáját jellemző paraméterek számát.

1. A Galilei-csoportok alapfogalmai

Ebben a fejezetben a Galilei-csoportokkal és azokhoz kapcsolódó alapfo-
galmakkal fogunk megismerkedni, melyek elengedhetetlenül szükségesek tár-
gyalásukhoz, mint például a topológiával, a Lie-algebrával, a féldirekt szor-
zattal, a kommutátor kociklusokkal vagy éppen azok gyenge kohomológiájá-
val.

1.1. Topologikus csoportok és Lie-csoportok

Először a topológiával kezdjük ismerkedésünket, ugyanis látni fogjuk,
hogy a Galilei-csoportok természetes módon azonosíthatók egy topologikus
csoporttal.

1. Definíció. A (T, τ)-t topologikus térnek nevezzük, ahol T egy nemüres
halmaz, τ pedig T részhalmazainak egy rendszere úgy, hogy a következők tel-
jesülnek.

1. A τ tartalmazza T -t és az üres halmazt,

2. τ bármely két elemének metszete is τ -ban van,

3. τ tetszőlegesen sok elemének uniója is τ -beli.

A τ elemeit nyílt halmazoknak, az egész τ halmazt topológiának nevezzük. A
nyílt halmazok T -re vonatkozó komplementerei a zárt halmazok. Azt mond-
juk, hogy a V ⊆ T halmaz környezete az x ∈ T pontnak, ha létezik olyan
U ⊆ T nyílt halmaz, melyre x ∈ U ⊆ V teljesül.

A féldirekt szorzat bevezethetőségéhez szükségünk lesz a topologikus
téren értelmezett függvények folytonosságának definiálásához.

2. Definíció. Ha (T1, τ1) és (T2, τ2) topologikus terek, akkor az f : T1 → T2

függvényt folytonosnak nevezzük, ha minden τ2-beli halmaz ősképe τ1-beli,
azaz nyílt halmaz ősképe nyílt.

Topologikus bázisokat a topologikus terek szorzata miatt fogjuk definiál-
ni.

3. Definíció. A (T, τ) topologikus téren B-t topologikus bázisnak nevezzük,
ha teljesíti a következő feltételeket.

1. Minden X ∈ B esetén X ∈ τ ,

2. valamint minden X ∈ τ -hoz létezik olyan {Xi}i∈I ∈ BI , hogy
X =

∪
i∈I Xi.

A topologikus terek szorzata a Galilei-csoportok definíciójának egyik leg-
fontosabb kelléke. A Galilei-csoportokat különböző topologikus terek szorza-
taként, és azokon ható műveletekkel definiáljuk.

1. Tétel. Legyen (T1, τ1) és (T2, τ2) topologikus tér, valamint pri (i = 1, 2)
jelölje a T1 × T2 → Ti kanonikus projekciót. Ha

B := {pr−1
1 (U) ∩ pr−1

2 (V ) : U ∈ τ1, V ∈ τ2},

akkor
τ := {U ⊆ T1 × T2 : ∀x ∈ U∃Ω ∈ B x ∈ Ω ⊆ U}

olyan topológiát definiál a T1×T2 halmazon, melynek B egy topologikus bázisa.

4. Definíció. A (T1, τ1), (T2, τ2) topologikus terek topologikus szorzatán azt a
(T, τ) topologikus teret értjük, melyre T = T1×T2, valamint τ a fenti tételben
megadott topológia.

A topologikus terek lokális kompaktságának definícióját szempontunkból
a speciálisan lokálisan kompakt féldirekt szorzatok meghatározása motiválja.

5. Definíció. Legyen (T, τ) topologikus tér.

1. A K ⊆ T részhalmazt kompaktnak hívjuk, ha minden nyílt fedésének
van véges befedő rendszere is. A K = T esetben a T teret nevezzük
kompaktnak.

2. Azt mondjuk, hogy a T tér lokálisan kompakt, ha minden pontjának
létezik kompakt környezete.

A Lie-algebrák bevezetése a kommutátor kociklusok definíciójának fontos
kelléke, és így nélkülözhetetlen a Galilei-csoportok tárgyalásában.

6. Definíció. (Lie-algebra) Adott egy V vektortér valamely F test felett és
egy kétváltozós művelet [., .] úgy, hogy

[., .] : V × V → V,

melyet kommutátornak vagy Lie-zárójelnek is neveznek és eleget tesz a követ-
kező tulajdonságoknak:

Bilineáris, ha

[ax+ by, z] = a[x, z] + b[y, z], [z, ax+ by] = a[z, x] + b[z, y]

minden a, b skalárra és minden x, y, z vektorra.
Antikommutatív vagy ferdén szimmetrikus, ha

[x, y] = −[y, x]

minden x, y vektorra.
Ha F karakterisztikája kettő, akkor egy szigorúbb feltételre is szükség van,

azaz akkor szükséges., hogy [x, x] = 0 minden x ∈ V vektorra teljesüljön.
Zárt vagyis teljesíti a Jacobi-azonosságot, ha

[x, [y, z]] + [y, [z, x]] + [z, [x, y]] = 0

minden x, y, z vektorra.

1.2. Féldirekt szorzat

A féldirekt szorzásra azért van szükségünk, mert ezzel a Galilei-csoport
szerkezete jellemezhető, ugyanis mint látni fogjuk, a Galilei-csoport féldirekt

szorzat szerkezetű csoport. Valamint a háromdimenziós összefüggő euklideszi
csoport definíciójában lesz még szükségünk rá, mely a teljes Galilei-csoport
egy fundamentális részcsoportja.

7. Definíció. Legyen A és H csoport, és továbbá legyen

t : H → Aut(A) h 7→ th

csoporthomomorfizmus. A H és A csoport féldirekt szorzata a t leképezésre
nézve a G = H ×t A csoport, melynek elemei (h,a) párokból állnak, és a
csoportműveletet a

(h, a)(h′, a′) = (hh′, ath(a
′))

összefüggés adja meg. Ha A és H topologikus csoportok, és
t ∈ Hom(H,Aut(A)) olyan, hogy

t′ : H × A → A (h, a) 7→ th(a)

leképezés folytonos a H × A szorzattopológiában, akkor a H ×t A féldirekt
szorzatot a szorzattopológiával ellátva topologikus féldirekt szorzatnak nevez-
zük.

Ha A és H szeparált, lokálisan kompakt, második megszámlálhatósági axi-
ómának eleget tevő topologikus csoport, valamint A kommutatív, akkor speci-
ális lokálisan kompakt féldirekt szorzatról beszélünk.

1.3. Galilei-csoportok

A teljes Galilei-csoport megismerése előtt bevezetünk néhány egyszerűbb,
speciálisabb csoportot, mely egy fajta felvezetést ad a Galilei-csoportok
tárgyalásának.

8. Definíció. 1. A Gs speciális Galilei-csoport izomorf és homeomorf az
R3 csoporttal. Elemeit "sebességként" interpretáljuk, és a T3(R) jelölést
használjuk rá.

2. Legyen E+ := SO3×κR3, ahol κ : SO3 → Aut(R3), valamint, ha min-
den R ∈ SO3 esetén teljesül, hogy κR : u 7→ Ru, akkor E+ csoportot
háromdimenziós összefüggő euklideszi csoportnak hívjuk, ahol a csoport
szorzás a következő:

(R′,u′)(R,u) := (R′R,u′ +R′u).

3. A Ge egyszerű Galilei-csoport alaphalmaza SO3×Gs×R3, és a rajta
értelmezett szorzás

(R′, v′,u′)(R, v,u) := (R′R, v′ +R′v,u′ +R′u).

4. A Gt teljes Galilei-csoport alaphalmaza SO3×R × Gs ×R3, a szorzás
pedig:

(R, τ, v,u)(R′, τ ′, v′,u′) = (RR′, τ + τ ′, Rv′ + v, Ru′ + u+ τ ′v).

Ezzel bevezettük a teljes Galilei-csoportot, melyre a következőekben, mint
ahogy eddig is tettük, csupán Galilei-csoportként fogunk hivatkozni.

1. Megjegyzés. A teljes Galilei-csoport részcsoportjaként tartalmazza az
egyszerű Galilei-csoportot, valamint T1(R)-et, T3(R)-at, Gs-et és SO3-at; ezt
a négy csoportot a teljes Galilei-csoport fundamentális részcsoportjának ne-
vezzük.

2. Megjegyzés. A teljes Galilei-csoport (R, τ, v,u) elemét felfoghatjuk R×R3

olyan transzformációjaként, amely

t 7→ t+ τ (1)

x 7→ Rx+ tv+ u (2)

formán hat. Valójában ez a transzformáció vezetett a Galilei-csoportok defi-
níciójához, ahol R×R3 a fizikai időt és teret jelképezik. T3(R) és SO3 elemeit,

mint R3 térbeli eltolásainak illetve térbeli forgatásoknak szokás nevezni. Gs

elemei olyan transzformációknak tekinthetőek, amelyek R3-ben egymáshoz ké-
pest állandó sebességgel mozgó koordináta-rendszereket viszonyítanak
egymáshoz, és T1(R) mint Gt részcsoportja az időeltolás csoportja.

A következő fejezetben az n = 1; 2 és a 3 térdimenziós Galilei-csoportokkal
fogunk mélyebben foglalkozni, ahol

G ∼= SOn ×R × Rn × Rn.

A szorzási szabály pedig R,R′ ∈ SOn, azaz R és R′ a kétdimenziós térbeli
esetben egy síkbeli forgatást, a háromdimenziós esetben pedig egy térbeli
forgatást reprezentál, az egydimenziós esetben ilyen nincs, valamint τ, τ ′ ∈ R

egy időbeli eltolást, továbbá v,v′ ∈ Tn(R) adott sebességű inerciarendszerek
közötti transzformációt, míg u,u′ ∈ Rn pedig egy térbeli/síkbeli/egyenesbeli
eltolást reprezentál, melyek esetén

(R, τ,v,u)(R′, τ ′,v′,u′) = (RR′, τ + τ ′, Rv′ + v, Ru′ + u + τ ′v)

alakú. A csoport egy (R, τ,v,u) elemének az inverze

(R, τ,v,u)−1 = (R−1,−τ,−R−1v, R−1(tv − u))

lesz. A csoport féldirekt szorzat szerkezetű; legyen A = {(1, τ,0,u)} ⊂ G

és H = {(R, 0,v,0)} ⊂ G. Ekkor t : H → Aut(A) a H csoporthatása az A

csoporton
t(R,0,v,0)(1, τ,0,u) = (1, τ,0, Ru + τv)

alakú lesz.
A csoport felírható egy (n+2)×(n+2) mátrix-csoport részcsoportjaként.

Ugyanis, ha R ∈ SOn, τ ∈ R, v ∈ Rn és u ∈ Rn, akkor ezeket
R v u

0 1 τ

0 0 1

 (3)

módon mátrixba rendezve, és ezt ellátva a szokásos mátrixszorzással éppen
a Galilei-csoporttal izomorf és homeomorf csoportot kapunk. Az így felírt
mátrixszal pedig már könnyen láthatóak, az előzőekben már felírt tér-idő
transzformációk, ahol t ∈ R és x ∈ Rn esetén (4) az őket reprezentáló

x

t

1

 (4)

vektorral való jobbról szorzással. Mely vektoron az iménti (3) mátrixszal való
szorzás a következőképpen hat:

Rx + vt+ u

t+ τ

1

 .

Tehát az (1) és (2) valóban fennáll.
A Galilei-csoport topologikus csoport, hiszen topológiát ad meg rajta az

Rk, ahol k =
(
n
2

)
+n+n+1, azaz egy dimenzióban R3, kétdimenziós esetben

R6, a háromdimenziós esetben pedig R10 szokásos topológiája. Továbbá Lie-
csoport is, hiszen a szorzás és az inverz-képzés is differenciálható.

1.4. Egyparaméteres részcsoportok

Ebben az alfejezetben az egyparaméteres részcsoportokat ismertetjük, és
néhány egyszerű példával szemléltetjük őket.

9. Definíció. Legyen T > 0 tetszőlegesen választott valós szám. Egy

R ⊃ (−T, T ) → N ⊂ G, t 7→ r(t), r(0) = e

leképezést a csoport e egységelemén áthaladó görbének nevezzük, ha

ρ := ϕ ◦ r

differenciálható. A görbe egyparaméteres részcsoport, ha

r(t)′r(s) = r(t+ s),

illetve
F (ρ(t), ρ(s)) = ρ(t+ s)

(−T < t, s, t+ s < T ).

A továbbiakban azt az esetet vizsgáljuk, amikor G Lie-csoport. Ekkor az
egységelemnek létezik olyan U nyílt környezete, mely homeomorf az Rn tér
egy Ω nyílt részhalmazával. Jelölje φ az U → Ω homeomorfizmust. Az r

egyparaméteres részcsoport φ-érintőjének az

a :=
d φ(r(t))

d t

∣∣∣
t=0

mennyiséget hívjuk. Jelölje g az egyparaméteres részcsoportok érintőinek a
halmazát. A g halmazon az alábbi műveleteket vezetjük be.

1. Ha a, b ∈ g az r és s egyparaméteres részcsoport φ-érintői, akkor legyen

a+ b :=
d φ(r(t)s(t))

d t

∣∣∣
t=0

.

2. Ha a ∈ g az r egyparaméteres részcsoport φ-érintője és λ ∈ R, akkor
legyen

λa :=
d φ(r(λt))

d t

∣∣∣
t=0

.

Ezekkel a műveletekkel g vektortér.
Megmutatható, hogy ha r és s egyparaméteres részcsoport, akkor létezik

a
lim
t→0

1

t2
φ(r(t)s(t)r(t)−1s(t)−1)

határérték, melyet egy újabb művelet, a kommutátor bevezetéséhez haszná-
lunk.

3. Ha a, b ∈ g az r és s egyparaméteres részcsoport φ-érintői, akkor legyen

[a, b] := lim
t→0

1

t2
φ(r(t)s(t)r(t)−1s(t)−1).

Igazolható, hogy g a fenti műveletekkel ellátva Lie-algebra, melyet a G

Lie-csoport Lie-algebrájának nevezünk.
Az előzőekben felírt (3) mátrixos alak segít hozzá, hogy könnyen megta-

lálhassuk a csoport Lie-algebrájának egy bázisát és azoknak kommutátorait.
Szemléltetésképpen felírjuk a 2+1 dimenziós Galilei-csoport egyparaméteres
részcsoportjait.

r1(t)=


cos t − sin t 0 0

sin t cos t 0 0

0 0 1 0

0 0 0 1

 r2(t)=


1 0 t 0

0 1 0 0

0 0 1 0

0 0 0 1

 r3(t)=


1 0 0 0

0 1 t 0

0 0 1 0

0 0 0 1



r4(t)=


1 0 0 t

0 1 0 0

0 0 1 0

0 0 0 1

 r5(t)=


1 0 0 0

0 1 0 t

0 0 1 0

0 0 0 1

 r6(t)=


1 0 0 0

0 1 0 0

0 0 1 t

0 0 0 1

 (5)

Amik pedig teljesítik az r(t1 + t2) = r(t1)r(t2) feltételt.
Innen pedig könnyen megadhatók az egyparaméteres részcsoportok érin-

tői. Tehát a továbbiakban legyen Ak (a síkbeli esetben (k = 1) a térbeli
esetben pedig (k = 1, 2, 3)) az SOn, valamint Bk (k ∈ [n]) a Tn tengelyirányú

egyparaméteres részcsoportjának érintője. Jelölje Dk (k ∈ [n]) ugyanarra a
koordináta-rendszerre vonatkoztatva Gs tengelyirányú egyparaméteres rész-
csoportjainak érintőit, és legyen végül F a T1 részcsoport érintője; minden
esetben a paraméterezést természetes módon választhatjuk.

Tehát az érintők, azaz a generátorok a 2+1 dimenziós esetben a követke-
zők lesznek:

A =


0 −1 0 0

1 0 0 0

0 0 0 0

0 0 0 0

B1 =


0 0 1 0

0 0 0 0

0 0 0 0

0 0 0 0

B2 =


0 0 0 0

0 0 1 0

0 0 0 0

0 0 0 0

 (6)

D1 =


0 0 0 1

0 0 0 0

0 0 0 0

0 0 0 0

D2 =


0 0 0 0

0 0 0 1

0 0 0 0

0 0 0 0

F =


0 0 0 0

0 0 0 0

0 0 0 1

0 0 0 0

 . (7)

1.5. Kommutátor kociklusok és gyenge kohomológiájuk

Most a kommutátor kociklusokat és a köztük ható gyenge kohomológiát
fogjuk bemutatni.

10. Definíció. Legyen g egy R feletti Lie-algebra a [., .] kommutátorral. Egy
κ : g × g → R bilineáris, antiszimmetrikus zárt függvényt a g Lie-algebra
kommutátor-kociklusának nevezünk. A κ és κ′ kommutátor-kociklusok koho-
mológok egymással, ha létezik olyan η : g → R lineáris függvény, mellyel

κ′(., .) = κ(., .) + η ◦ [., .].

A két kommutátor-kociklus gyengén kohomológ, ha κ és κ′ vagy −κ és κ′

kohomológ egymással.

Kommutátor táblázatnak nevezünk egy olyan táblázatot, melynek elemei
a csoport generátoraink kommutátorai, számszorosai vagy lineáris kombi-
nációi. Mindezt oly módon, hogy az ij-edik helyre az i-edik és a j-edik



1. A GALILEI-CSOPORTOK ALAPFOGALMAI 12

generátor kommutátora által kapott generátort, vagy azok lineáris kombi-
nációit írjuk. Innen már könnyen felírhatjuk egy tetszőleges κ kommutá-
tor kociklus paraméteres táblázatát, mely teljesíti az antiszimmetriát (tehát
κ(a, b) = −κ(b, a)), a bilinearitást (azaz mindkét változójában lineáris), és a
zártságot, vagyis a Jacobi-azonosságot, azaz

κ([a, b], c) + κ([c, a], b) + κ([b, c], a) = 0

minden a, b, c generátorra.

1.6. Speciális lineáris csoportok

Ebben a részben némi kitekintést teszünk a Galilei-csoportok elméletétől.
Azaz egyfajta bevezetésképpen vizsgálat alá vetjük egy másik Lie-csoport
kohomológia osztályait. Méghozzá a speciális lineáris csoportot, melyet a
következő definícióval vezetünk be.

11. Definíció. Az SL(2,R) :=

{
M =

(
a b

c d

)
: detM = 1; a, b, c, d ∈ R

}
halmazt speciális lineáris csoportnak nevezzük.

Ez valóban csoportot alkot, hiszen, ha detA = 1 és detB = 1, akkor
detAB = detA · detB = 1 · 1 = 1. Egységeleme pedig az egységmátrix. És
mivel determinánsuk nem 0, létezik inverzük, melyre detA−1 = (detA)−1 =

1−1 = 1, így az inverzük is SL(2,R)-beli.
Továbbá egyparaméteres részcsoportjai:

r1 :=

(
1 t

0 1

)
r2 :=

(
1 0

t 1

)
r3 :=

(
t 0

0 1
t

)
.

Ezek valóban SL(2,R) elemei, hiszen mindegyikük determinánsa 1, továb-
bá függetlenek is, és egy 3 dimenziós alteret generálnak a 4 dimenziós 2×2-es
valós mátrixok terén, tehát valóban kifeszítik SL(2,R)-t, hiszen ha SL(2,R)

4 dimenziós lenne, akkor minden 2× 2-es mátrix determinánsa 1 lenne, ami
nyilván valóan nem igaz.

Az egyparaméteres részcsoportokból számított bázis pedig:

a :=

(
0 1

0 0

)
b :=

(
0 0

1 0

)
c :=

(
1 0

0 −1

)
.

Ezeket az egyparaméteres részcsoportok origóban vett deriváltjaiból szá-
mítottunk. Továbbá a kommutátor táblázatuk a következőnek adódik,

0 c −2a

−c 0 2b

2a −2b 0

 .

Egy tetszőleges κ kommutátor kociklusuk paraméteres táblázata pedig a
következő lesz: 

0 m1 m2

−m1 0 m3

−m2 −m3 0

 .

Most pedig megnézzük, hogy a zártsági kritériummal/Jacobi-azonosság-
gal egyszerűsíthető-e az előbbi paraméteres táblázat. Mivel egy elég kicsi
mátrix-szal rendelkezünk, ezt most szemléltetésként, kézileg fogjuk végigszá-
molni. Vagyis végig fogjuk nézni, hogy minden esetere tejesül-e a zártság.
Összesen 3 ·3 ·3, azaz 27 eset van. Ha két azonos elem is szerepel azok között,
melyekre a Jacobi-azonosságot felírtuk, akkor

κ([x, x], y) + κ([y, x], x) + κ([x, y], x) = κ(0, y)− κ([x, y], x) + κ([x, y], x) ≡ 0.

Tehát csak azokat az eseteket kell megvizsgálnunk, ahol mind a 3 elem
különböző, ilyen pedig 3 · 2 · 1, azaz 6 eset van, sőt mivel ezek közül 3-3
egymás permutációja, így csak az alább részletezett 2 eset marad. Ahol
pedig felhasználtuk, hogy κ antiszimmetrikus, tehát κ(x, x) ≡ 0 és κ(a, b) +

κ(b, a) ≡ 0.

κ([a, b], c) + κ([c, a], b) + κ([b, c], a) = κ(c, c) + 2κ(a, b) + 2κ(b, a) ≡ 0

κ([b, a], c) + κ([c, b], a) + κ([a, c], b) = −κ(c, c)− 2κ(b, a)− 2κ(a, b) ≡ 0

Tehát a változók táblája nem változik. Így pedig a κ kommutátor kociklus
gyengén kohomológ egy κ′-vel, melyet a következőképpen adhatunk meg:

κ′(a, b) = κ(a, b) + η(c) = m1 + η(c) ⇒ η(c) := −m1

κ′(a, c) = κ(a, c) + η(−2a) = m2 − 2η(a) ⇒ η(a) :=
m2

2

κ′(b, c) = κ(b, c) + η(2b) = m3 + 2η(b) ⇒ η(b) := −m3

2
.

Azaz létezik olyan κ′ kommutátor kociklus, melynek változóiban nem
marad paraméter. Pontosabban olyan κ′, hogy κ′(x, y) ≡ 0 minden x, y-
ra.

2. Alacsony dimenziós Galilei-csoportok

Ebben a fejezetben olyan kis dimenziós Galilei-csoportokkal fogunk fog-
lalkozni, mint az 1 + 1, 2 + 1 és a 3 + 1 dimenziós Galilei-csoportok.

2.1. Galilei-csoport 1+1 dimenziós téridőn

Ebben a részben először néhány kisebb Galilei-csoport kohomológia osz-
tályait fogjuk megvizsgálni, melyek eleinte még akár kézileg is számolhatóak,
de a dimenzió növekedtével egyre nagyobb szükségünk lesz gépi programok
alkalmazására is. Kezdjük tehát az 1+1 dimenziós Galilei-csoporttal.

Tehát most néhány egyszerű számolást fogunk bemutatni. Az 1+1 di-
menziós esetben egyetlen forgatás létezik, ami a helyben hagyás, így ez a
csoport a következő mátrixszal reprezentálható:

1 v u

0 1 τ

0 0 1

 ,

ahol v, u és τ valós paraméter értékek. Az egyparaméteres részcsoportjai
pedig a következők:

r1(v) :=


1 v 0

0 1 0

0 0 1

 r2(u) :=


1 0 u

0 1 0

0 0 1

 r3(τ) :=


1 0 0

0 1 τ

0 0 1

 .

Az egyparaméteres részcsoportok érintői, melyek a csoport generátorai

B =


0 1 0

0 0 0

0 0 0

 D =


0 0 1

0 0 0

0 0 0

 F =


0 0 0

0 0 1

0 0 0

 .

Az itt felírt bázisok kommutátorait egy táblázatba foglaltuk, melyet
Mapel 18 programmal számoltunk ki, és az alapján írtuk fel, hogy a táb-

lázatba az (i, j)-edik helyre 0-t írtunk, ha a kommutátoruk nullmátrix volt, k-
t, ha a kommutátoruk a k-adik báziselem volt, és −k-t, ha a k-adik báziselem
−1-szerese volt a kommutátoruk, és így a következő (8) táblázat adódott,
melyet a következőben kommutátor táblának fogunk hívni. A kommutátoraik
[B,D] = 0, [B,F ] = D valamint [D,F ] = 0 voltak, ahonnan a kommutátor
táblázatuk pedig 

0 0 2

0 0 0

−2 0 0

 , (8)

ahol a bázis elemeket az "abc" szerinti sorrendjük alapján indexeltük, tehát
e1 = B, e2 := D, e3 := F megfeleltetést alkalmaztuk. Az ei elemekre pedig
az i indexszel hivatkoztunk.

Továbbá a κ paraméter táblázata (9), ahol egyelőre azt használtuk ki,
hogy a κ antiszimmetrikus, így (j, i) helyen szereplő elem éppen −1-szerese
az (i, j)-ediknek, továbbá éppen emiatt a főátlóbeli elemek mind nullák.

0 a1 a2

−a1 0 a3

−a2 −a3 0

 (9)

Innen pedig az a2-es elem könnyen kiejthető (10) módon. Mely egy a
κ-val gyengén kohomológ κ′ megadásával tehető meg.

κ′(B,F ) = κ(B,F ) + η(D) = a2 + η(D) ⇒ η(D) := −a2 (10)

Most definiálhatunk egy κ′ paraméter táblázatot úgy, hogy
0 m1 0

−m1 0 m2

0 −m2 0

 . (11)

Melyre a Jacobi-azonosság minden tényezője azonosan 0 lesz. Tehát az
így kapott κ′ teljesíti a kommutátor kociklusok minden tulajdonságát. És

nem adható meg más olyan kommutátor kociklus, mely κ-val gyengén koho-
mológ és kevesebb paramétert használ κ′-nél. Azaz az 1+1 dimenziós Galilei-
csoportnál 2 paraméterünk fog megmaradni.

2.2. Galilei-csoport 2+1 dimenziós téridőn

A 2+1 dimenziós Galilei-csoport egyparaméteres részcsoportjait (5) már
megadtuk egyfajta szemléltetésképp az egyparaméteres részcsoportokról szó-
ló fejezetben. Valamint a generátorokat is megadtuk, melyek (6) és (7) voltak.
A kommutátor táblázatát Maple segítségével határoztuk meg a generátorok
mechanikus felírásával, majd ezeket két for ciklusba tettük, és a kommutáto-
rok eredményeit egy mátrixba tároltuk el az előző fejezethez hasonló módon.
Így pedig a (12) mátrixot kaptuk.

0 3 −2 5 −4 0

−3 0 0 0 0 4

2 0 0 0 0 5

−5 0 0 0 0 0

4 0 0 0 0 0

0 −4 −5 0 0 0


(12)

A bázis elemeket az "abc" szerinti sorrendjük, és az indexeik alapján állítot-
tuk sorba, azaz e1 = A, e2 := B1, e3 := B2, e4 := D1, e5 := D2, e6 := F

megfeleltetést alkalmaztuk. Az ei elemekre pedig az i indexszel hivatkoztunk.
Az imént már felvázolt kommutátor táblából Maple segítségével felírtuk

egy tetszőleges κ kommutátor kociklusának paraméteres táblázatát, mely
teljesíti az antiszimmetriát, a bilinearitást és a zártságot, így az alábbi tulaj-

donságokkal bíró, táblázatot kaptuk.

0 a1 a2 a3 a4 a5

−a1 0 a6 a7 0 −a4

−a2 −a6 0 0 a7 a3

−a3 −a7 0 0 0 0

−a4 0 −a7 0 0 0

−a5 a4 −a3 0 0 0


Továbbá a 2+1 dimenzióban a következőekben megadunk egy κ′ kommu-

tátor kociklust, mely gyengén kohomológ a κ-val, melyet a következőképpen
tehetünk meg.

κ′(e1, e2) = κ(e1, e2) + η(e3) = a1 + η(e3) = 0 ⇒ η(e3) := −a1 (13)

κ′(e1, e3) = κ(e1, e3) + η(−e2) = a2 − η(e2) = 0 ⇒ η(e2) := a2

κ′(e1, e4) = κ(e1, e4) + η(e5) = a3 + η(e5) = 0 ⇒ η(e5) := −a3

κ′(e1, e5) = κ(e1, e5) + η(−e4) = a4 − η(e4) = 0 ⇒ η(e4) := a4

κ′(e2, e6) = κ(e2, e6) + η(e4) = −a4 + η(e4) = 0 X

κ′(e3, e6) = κ(e3, e6) + η(e5) = a3 + η(e5) = 0 X

Az így definiált η-k mellett, ha a többi definiálatlan báziselemhez 0-t
rendelünk, az így kapott κ′ valóban gyengén kohomológ lesz κ-val és csupán
az a5, a6 és az a7 paraméterektől fog függni.

2.3. Galilei-csoport 3+1 dimenziós téridőn

A 3+1 dimenziós eset nagysága tökéletes példa arra, hogy már egy ilyen
kis dimenziós esetben is mekkora szükségünk van a Maple-re. A dimenzió
növekedtével pedig egyre nagyobb szükségünk lesz számítógépes programok

használatára, mivel kézileg egyre számolhatatlanabb lesz a Galilei-csoport.
Most pedig felvázoljuk a 3+1 dimenziós Galilei-csoport generátorait, me-

lyek a következőek.

A1 :=



0 0 0 0 0

0 0 −1 0 0

0 1 0 0 0

0 0 0 0 0

0 0 0 0 0


A2 :=



0 0 1 0 0

0 0 0 0 0

−1 0 0 0 0

0 0 0 0 0

0 0 0 0 0



A3 :=



0 −1 0 0 0

1 0 0 0 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0


B1 :=



0 0 0 1 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0



B2 :=



0 0 0 0 0

0 0 0 1 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0


B3 :=



0 0 0 0 0

0 0 0 0 0

0 0 0 1 0

0 0 0 0 0

0 0 0 0 0


D1 :=



0 0 0 0 1

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0



D2 :=



0 0 0 0 0

0 0 0 0 1

0 0 0 0 0

0 0 0 0 0

0 0 0 0 0


D3 :=



0 0 0 0 0

0 0 0 0 0

0 0 0 0 1

0 0 0 0 0

0 0 0 0 0


F :=



0 0 0 0 0

0 0 0 0 0

0 0 0 0 0

0 0 0 0 1

0 0 0 0 0


A 3 + 1 dimenziós Galilei-csoport kommutátor táblázata:

0 3 −2 0 6 −5 0 9 −8 0

−3 0 1 −6 0 4 −9 0 7 0

2 −1 0 5 −4 0 8 −7 0 0

0 6 −5 0 0 0 0 0 0 7

−6 0 4 0 0 0 0 0 0 8

5 −4 0 0 0 0 0 0 0 9

0 9 −8 0 0 0 0 0 0 0

−9 0 7 0 0 0 0 0 0 0

8 −7 0 0 0 0 0 0 0 0

0 0 0 −7 −8 −9 0 0 0 0



.

A κ paraméter táblázata, a zártság figyelembe vételével:

0 a1 a2 0 a4 a5 0 a7 a8 0

−a1 0 a10 −a4 0 a13 −a7 0 a16 0

−a2 −a10 0 −a5 −a13 0 −a8 −a16 0 0

0 a4 a5 0 0 0 a27 0 0 a16

−a4 0 a13 0 0 0 0 a27 0 −a8

−a5 −a13 0 0 0 0 0 0 a27 a7

0 a7 a8 −a27 0 0 0 0 0 0

−a7 0 a16 0 −a27 0 0 0 0 0

−a8 −a16 0 0 0 −a27 0 0 0 0

0 0 0 −a16 a8 −a7 0 0 0 0



.

Analóg módon a (13) számolással kaptunk (3+1) dimenzióban olyan κ′-t,
mely gyengén kohomológ a κ-val, ahol

η(e3) := −a1, η(e2) := a2, η(e6) := −a4, η(e5) := a5, η(e9) := −a7

η(e8) := a8, η(e1) := −a10, η(e4) := −a13, η(e9) := −a7, η(e7) := −a16.

A többi definiálatlan bázis elemre pedig η 0-t rendel, ekkor (3+1)-ben csupán
egyetlen paraméter marad a27, ami pedig a fizikai tömeg az [1] alapján.

3. Kohomológiaosztályok meghatározása

Maple 18 program használatával

Ebben a fejezetben be fogunk mutatni egy programot, amit azért írtunk,
hogy segítségével megtudjuk határozni magasabb dimenziós esetekben is a
Galilei-csoportok gyenge kohomológiáit meghatározó paraméterek számát,
oly esetekben is, amikor a kézi számolás már nem követhető módszer. S
mindehhez csupán a tér dimenziójának megadására lesz szükségünk.

Az alábbi programrészlettel megalkotjuk a generátorait a k + 1 téridő
dimenziós Galilei-csoportnak, melynek bemenete egy k szám, azaz a tér di-
menziója, kimenete pedig a generátorok egy listája. Melyet úgy oldottunk
meg, hogy írtunk egy procedúrát az előbbi bemeneti paraméterrel és a kö-
vetkező lokális változókkal: tt, ee, ff , e, ret.

Az első két for ciklussal a forgatások által generált mátrixokat adjuk meg.
Olyan módon, hogy a bal felső sarokban egy n×n-es antiszimmetrikus mát-
rixot töltünk ki, úgyhogy a főátló felett egy darab egyes van vagy egy darab
mínusz egyes, és ezt antiszimmetrikus módon tükrözzük a főátlóra, így meg-
kapva egy-egy generátort.

Ezeket a generátorokat abban a sorrendben alkotjuk meg, hogy a főátló
mellől az első sorból indulva, (azaz az (1, 2) indexről) és arra egy −1-est írva
balról jobbra haladunk a blokk mátrix határáig generátoronként úgy, hogy az
egymást követő generátorokon a −1 és a +1 alternálva követi egymást, majd
a blokk mátrix határához érve a következő sorból, a főátló mellől folytatjuk
a procedúrát, egészen addig, amíg végig nem érünk a teljes blokk mátrixon.

A harmadik for ciklussal a térbeli eltolások által generált bázisokat adjuk
meg, a szokásos sorrendben (fentről lefelé haladva az indexeken). A negye-
dik for ciklussal pedig az inercia rendszerek közti transzformáció és az idő
eltolásból származó generátort adtuk meg, szintén a szokásos sorrendben. S
ezeket az e vektorban tárolunk el. Majd pedig a ret-ben egy listát csinálunk
e-ből.

> with(LinearAlgebra):
> create:=proc (K::integer)

> local tt, ee, ff, e, ret:
> tt:=1:
> for ee to K-1 do
> for ff from ee+1 to K do
> e[tt]:=Matrix(K+2,K+2,0):
> e[tt][ee,ff]:=(-1)^(ee+ff):
> e[tt][ff,ee]:=-e[tt][ee,ff]:
> tt:=tt+1:
> end do:
> end do:
> for ee to K do
> e[tt]:=Matrix(K+2,K+2,0):
> e[tt][ee,K+1]:=1:
> tt:=tt+1:
> end do:
> for ee to K+1 do
> e[tt]:=Matrix(K+2,K+2,0):
> e[tt][ee,K+2]:=1:
> tt:=tt+1:
> end do:
> dim:=(1/2)*K^2+(3/2)*K+1:
> ret:=[seq(e[i],i=1..dim)]:
> return(ret):
> end proc:

Ebben a programrészletben a k+1 tér-idő dimenziós Galilei-csoport kom-
mutátor táblázatát fogjuk kiszámítani. Ahol a bemenetünk a tér dimenzió-
ja k és a generátorok listája lesz, kimenetünk pedig a kommutátor mátrix.
Melyre szintén procedúrát írtunk, az előbbi bementi paraméterrel és a dimv,
komm, ee, e, ff , kom, kk lokális változókkal. Ahol dimv a generátorok
dimenzióját adja meg.

Az első for ciklusban az e változóval felvesszük a ret listából a gene-
rátorokat. A második két for ciklusban pedig kiszámítjuk a generátorok
kommutátorait, azaz a kommutátor táblázatukat.

Ennek a kommutátor táblázatnak a kiszámításában érdekesség volt, hogy
mivel a Maple nem tud mátrixot mátrixszal összehasonlítani, ki kellett talál-

ni egy módszert, amivel könnyen és gyorsan össze tudjuk hasonlítani, hogy
két mátrix egyenlő-e. A legegyszerűbb ötletként az adódott, hogy kivonjuk
egymásból a két mátrixot és kiszámítjuk az így kapott mátrix Frobenius-
normáját, melyre a norma definíciójából következőleg pontosan akkor lesz 0,
ha a mátrix a nullmátrix, tehát a két mátrix megegyezik.

Egyébiránt ismeretes, hogy

∥A∥2F =
m∑
i=1

n∑
j=1

|aij| = Trace(AAT ).

Továbbá tudjuk, hogy a bázisok kommutátorai csakis bázisbeli elemek,
vagy azok −1-szeresei vagy 0-k lehetnek a Galilei-csoportok esetén. Így tehát,
ha az [ei, ej] = ek, akkor a táblázat (i, j)-edik elemét k-nak definiáljuk, ha
[ei, ej] = −ek, akkor pedig −k-nak; egyébiránt viszont 0-nak.

> commmatrix:=proc(K,ret)
> local dimv,komm,ee,e,ff,kom,kk:
> dimv:=(1/2)*K^2+(3/2)*K+1:
> komm:=Matrix(dimv,dimv,k):
> for ee to nops(ret) do
> e[ee]:=ret[ee]:
> end do:
> for ee to dimv do
> for ff to dimv do
> kom:=e[ee].e[ff]-e[ff].e[ee]:
> if 0 < Trace(kom.Transpose(kom)) then
> for kk to dimv do
> if Trace((kom-e[kk]).Transpose(kom-e[kk]))=0 then
> komm[ee,ff]:=kk:
> kk:=dimv:
> end if:
> if Trace((kom+e[kk]).Transpose(kom+e[kk]))=0 then
> komm[ee,ff]:=-kk:
> kk:=dimv:
> end if:
> end do:
> else komm[ee,ff]:=0:

> end if:
> end do:
> end do:
> return(komm):
> end proc:

Az alábbiakban egy valós paraméterekkel paraméterezett mátrixot de-
finiálunk, mely teljesíti az antiszimmetriai tulajdonságát, bilinearitását a
Galilei-csoport egy tetszőleges κ kommutátor kociklusának. Először csak
az antiszimmetriát adjuk meg, és azt, hogy a főátlóban csak csupa 0 lehet,
mely az antiszimmetria következménye. A bemenet a tér dimenzió, a kimenet
pedig az előbbi mátrix, és a paraméterek listája.

A κ paraméter táblázatának meg határozásához még a Jacobi-azonosság
teljesülésére is szükségünk lesz, ám ezt másik procedúrákkal fogjuk megolda-
ni.

> Mcreate:=proc(k)
> local dimv, numofparameters, alist, M, tt, ee, ff:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> numofparameters:=(1/2)*dimv*(dimv-1):
> alist:=[seq(a[i],i=1..numofparameters)]:
> M:=Matrix(dimv,dimv,0):
> tt:=1:
> for ee to dimv do
> for ff from ee+1 to dimv do
> M[ee,ff]:=alist[tt]:
> tt:=tt+1:
> M[ff,ee]:=-M[ee,ff]:
> end do:
> end do:
> return alist, M:
> end proc:

A zerocohom procedúra bemeneti értékei a tér dimenzió, a kommutátor
tábla, az előbbi antiszimmetrikus mátrix és a változók listája. Mely azt
csinálja, hogy végig megy egy a generátoroknak megfeleltethető számok listá-
ján és a kommutátor mátrix elemein, majd összehasonlítja azok abszolút
értékét. (Az előbbi megfeleltetést úgy kell érteni, hogy a generátorokat egy

számmal indexeltük, és a kommutátor táblázatban ezek a számok szerepeltek
a generátorok helyett.)

Ha egyezést talál, akkor ahol az egyezés történt, annak a paraméternek
az indexét eltárolja a zeroplace-ben, a mátrixban elfoglalt helyét pedig a
zeroplaceinmatrix-ban, majd pedig dimv-re állítja a mátrix elemeken futó
indexet. Így biztosítva azt, hogy minden generátorhoz legfeljebb egy kinul-
lázható index társuljon. Az így eltárolt indexek pedig a kimeneti értékek lesz-
nek, azaz a kimeneti értékeink azt adják meg, hogy a κ kommutátor kociklus
mely paraméterei nullázhatóak ki egy megfelelően megadott κ′ kommutátor
kociklussal, mely gyengén kohomológ κ-val.

> zerocohom:=proc(k,comm,MM,alist)
> local alistv, numofp, MMv, dimv, zeroplace,
> zeroplaceinmatrix, e, aa, bb:
> alistv:=alist:
> numofp:=nops(alistv):
> MMv:=MM:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> zeroplace:={}:
> zeroplaceinmatrix:={}:
> for e to dimv do
> for aa to dimv do
> for bb to dimv do
> if abs(comm[aa,bb])=e then
> zeroplace:=‘union‘(zeroplace,{op(1,MM[aa,bb])}):
> zeroplaceinmatrix:=‘union‘(zeroplaceinmatrix,{[aa,bb]}):
> aa:=dimv:
> bb:=dimv:
> end if:
> end do:
> end do:
> end do:
> return zeroplace, zeroplaceinmatrix:
> end proc:

Az MMcreate procedúra bemeneti értékei a tér dimenzió és a zero-
placesinmatrix, azaz a mátrix előbb megadott nullhelyei, melyek kinulláz-

hatóak egy megfelelően megadott κ′ gyengén kohomológ kommutátor kocik-
lussal. A procedúra újraindexeli a változók halmazát úgy, hogy újra írja a
változókat 1-től egészen az eredeti paraméter szám, mínusz a nullparaméte-
rek száma. Majd definiálunk egy n× n-es nullmátrixot, amit újra feltöltünk
változókkal, de úgy, hogy azokat a helyeket kihagyjuk a feltöltéskor, amik az
előbb meghatározott kinullázható paraméterek voltak. Továbbá figyelembe
vesszük a κ′ antiszimmetriáját. Majd visszatérünk a változók új listájával és
az újradefiniált κ′ paraméter táblázatával.

> MMcreate:=proc(k,zeroplacesinmatrix)
> local dimv, numofp, zeroplaceinmatrix, zeronum, alistv,
> MMv, tt, ee, ff:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> numofp:=(1/2)*dimv*(dimv-1):
> zeroplaceinmatrix:=zeroplacesinmatrix:
> zeronum:=nops(zeroplaceinmatrix):
> alistv:=[seq(a[i],i=1..numofp-zeronum)]:
> MMv:=Matrix(dimv,dimv,0):
> tt:=1:
> for ee to dimv do
> for ff from ee+1 to dimv do
> if member([ee,ff],zeroplaceinmatrix) then
> else MMv[ee,ff]:=alistv[tt]:
> MMv[ff,ee]:=-MMv[ee,ff]:
> tt:=tt+1:
> end if:
> end do:
> end do:
> return alistv, MMv:
> end proc:

A most következő zerosinM procedúra bemenete a tér dimenzió, egy
mátrix, és nullindexek egy listája, melyben egy dupla for ciklussal megnézzük,
hogy a mátrix elemek közül melyek szerepelnek a nullindex listában. Ha egy
mátrix elem megtalálható a zeroindex-ben, akkor eltároljuk a zeroinmat

listában, mely a kimeneti értékünk is lesz.

> zerosinM:=proc(k,Mn,zeroindex)
> local dimv, zeroinmat, ee, ff:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> zeroinmat:={}:
> for ee to dimv do
> for ff from ee+1 to dimv do
> if member(op(1,Mn[ee,ff]),zeroindex) then
> zeroinmat:=‘union‘(zeroinmat,[ee,ff]):
> end if:
> end do:
> end do:
> return zeroinmat:
> end proc:

Most pedig egyenletrendszerek egy halmazát definiáljuk az előbbi táblázat
paramétereire, mellyel célunk, hogy teljesítsék a Jacobi-azonosságot. Ezt úgy
tesszük meg, hogy egymásba ágyazunk 3 db for ciklust, és a ciklus elején
mindig kinullázunk egy S paramétert, mely az új egyenletek hozzávételében
fog minket segíteni. És azt mondjuk, hogy ha az ij-edik kommutátor nem
nulla, akkor az S értékét növeljük κ([ei, ej], ek)-val, mindezt tekintve a Jacobi-
azonosság szerinti 3 db κ-val, azaz az ijk, jki, kij indexekre megvizsgált
κ-kra.

Ha pedig az így kapott S és annak mínusz egyszerese sem szerepel az
egyenletrendszerünkben a ciklus végén, akkor S-t felvesszük az egyenletrend-
szerbe és folytatjuk tovább a ciklust. Így pedig a procedúra visszatérési értéke
a Jacobi-azonosságból származó egyenletek listája lesz.

> setofeq:=proc(k,komm,M)
> local dimv, eqset, ee, ff, kk, S:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> eqset:=:
> for ee to dimv do
> for ff from ee+1 to dimv do
> for kk from ff+1 to dimv do
> S:=0:
> if komm[ee,ff]<>0 then
> S:=S+signum(komm[ee,ff])*M[abs(komm[ee,ff]),kk]:

> end if:
> if komm[ff,kk]<>0 then
> S:=S+signum(komm[ff,kk])*M[abs(komm[ff,kk]),ee]:
> end if:
> if komm[kk,ee]<>0 then
> S:=S+signum(komm[kk,ee])*M[abs(komm[kk,ee]),ff]:
> end if:
> if S<>0 then
> if member(S,eqset) or member(-S,eqset) then
> else eqset:=‘union‘(eqset,S = 0):
> end if:
> end if:
> end do:
> end do:
> end do:
> return eqset:
> end proc:

Az MMcreate κ′ paramétereire a Jacobi-azonosságot alkalmazva egy
olyan egyenletrendszert kapunk, melyben minden egyenlet esetében κ′-nek
egy, kettő vagy három paramétere lesz nullával egyenlő, méghozzá úgy, hogy
a paraméterek a bal oldalon helyezkednek el, a jobb oldal pedig nulla lesz.

A simpeqset procedúrával azokat a paramétereket fogjuk megadni, ahol
az egyenlet csupán egy paraméterből állt, azaz ±ai = 0 alakú. A procedúra
egyetlen bemenete egy egyenletrendszer lesz. Első lépésben, ha az egyenlet-
rendszerünkben többször szerepel egy egyenlet vagy annak konstansszorosa,
mely esetünkben csak mínusz egyszerese lehet, akkor a feleslegesen megjelenő
többszörös egyenletektől megszabadulunk.

Ezt követően egy for ciklus segítségével végig megyünk az így kapott
egyenletek listáján, és leellenőrizzük, hogy az egyenlet jobb oldala valóban
nulla, és hogy a baloldal hossza egy, tehát ai = 0 alakú, akkor az i inde-
xet eltároljuk a temp1null-ban, majd töröljük az egyenletet az egyenletek
listájáról. Valamint még azt az esetet vizsgáljuk meg, hogy ha a baloldal
hossza kettő, azaz −ai = 0 alakú az egyenletünk (mivel a Maple az előjelet,
műveleteket, mint az összeadás, szorzás is az egyenlet hosszának számolja),

akkor hasonlóan elmentjük az adott i indexét a paraméternek, majd töröljük
az egyenletek közül.

A visszatérési értékeink az így elmentett paraméterek indexeinek listája,
valamint ezen egyenletek elhagyásával kapott egymástól független egyenle-
tekből álló egyenletrendszer lesz.

> simpeqset:=proc(eqset)
> local eqsetv, cce, tempcc, temp1null, lengthrightside,
> lengthleftside, leftindex:
> eqsetv:=eqset:
> for cce in eqsetv do
> if rhs(cce)=0 and member(-lhs(cce)=0,eqsetv) then
> eqsetv:=‘minus‘(eqsetv,cce) end if end do:
> tempcc:=eqsetv:
> temp1null:={}:
> for cce in tempcc do
> lengthrightside:=nops(rhs(cce)):
> lengthleftside:=nops(lhs(cce)):
> if rhs(cce)=0 and lengthleftside=1 then
> leftindex:=op(1,lhs(cce)):
> temp1null:=‘union‘(temp1null,leftindex):
> tempcc:=‘minus‘(tempcc,cce) end if:
> if rhs(cce)=0 and lengthleftside=2 and op(1,lhs(cce))=-1
> then
> leftindex:=op(1,op(2,lhs(cce))):
> temp1null:=‘union‘(temp1null,leftindex):
> tempcc:=‘minus‘(tempcc,cce)
> end if:
> end do:
> return temp1null, tempcc:
> end proc:

A dimker egy olyan mátrixot reprezentál, melynek soraiban az egyenlet-
rendszerek, oszlopaiban a változók állnak, ez LL, melyet úgy reprezentáltunk,
hogy először létre hoztunk egy nullmátrixot, melynek annyi sora van, ahány
egyenlet, és annyi oszlopa, ahány változó. Ezt követően pedig leellenőriztük,
hogy az egyenletrendszerünk valóban nem volt hibásan kitűzve, azaz a jobb

oldalon valóban minden nulla, ha mégsem volt helyes a kitűzés, akkor hiba
üzenettel térünk vissza.

Aztán egy for ciklus segítségével feltöltöttük a mátrixot úgy, hogy a meg-
felelő indexek helyére a megfelelő konstans kerüljön az egyenletek és a vál-
tozók összefüggésében. Továbbá ez teljesíti, hogy LL · x = 0. A dimenzió
tételből pedig tudjuk, hogy

dim(Ker(LL)) + dim(Ran(LL)) = p,

ahol p a változók, azaz a paraméterek száma. Tehát

dim(Ker(LL)) = p− dim(Ran(LL)).

Így pedig dim(Ker(LL))-lel azt kaptuk meg, hogy a κ′ kommutátor ko-
ciklusban hány függő paraméterünk lesz. Azaz, hogy az n tér dimenziós
Galilei-csoport gyenge kohomológia osztályai között hány függő paraméter
van.

> dimker:=proc(eqset,varset)
> local eqsetfinal, numofeq, numofparameters, LL, tt, cce,
> e, dimensionofker:
> eqsetfinal:=eqset:
> numofeq:=nops(eqsetfinal):
> numofparameters:=nops(varset):
> LL:=Matrix(numofeq,numofparameters,0):
> tt:=1:
> for cce in eqsetfinal do
> if rhs(cce)<>0 then print(hiba)
> end if:
> for e to numofparameters do
> LL[tt,e]:=coeff(lhs(cce),a[e]):
> end do:
> tt:=tt+1:
> end do:
> dimensionofker:=numofparameters-Rank(LL):
> return dimensionofker:
> end proc:

Ezek után már pedig csak le kell futtatnunk a programunkat a pro-
cedúráknak az iméntiekben vázolt egymás után történő végrehajtásukkal,
melynek használata az alábbiakban figyelhető meg a 2 + 1 tér-idő dimen-
ziós Galilei-csoport kommutátor kociklusaihoz tartozó gyenge kohomológia
osztályok végig számolásával.

> k:=2:
> dimv:=(1/2)*k^2+3*k*(1/2)+1:
> numofparameters:=(1/2)*dimv*(dimv-1):
> generators:=create(k):
> comm:=commmatrix(k,generators):
> print("Az MM mátrix elsö megalkotása,
> és a paraméterek listája."):
> cc:=Mcreate(k):
> alist:=cc[1]:
> MM:=cc[2]:
> print("A null kohomológia a következö paramétereket
> adja 0-nak."):
> cc:=zerocohom(k,comm,MM,alist):
> zeroplacesinmatrix:=cc[2]:
> print("A módosított paraméterkészlet és az MM mátrix,
> az elözöekkel."):
> cc:=MMcreate(k,zeroplacesinmatrix):
> alist:=cc[1]:
> MM:=cc[2]:
> print("A Jacobi-azonosság a következö egyenleteket
> adja"):
> eqset:=setofeq(k,comm,MM):
> print("A null paraméterek és a null paraméterekkel
> egyszerüsített egyenletek."):
> cc:=simpeqset(eqset):
> zeroindex:=cc[1]:
> newzeroplacesinmatrix:=zerosinM(k,MM,zeroindex):
> allzeros:=‘union‘(zeroplacesinmatrix,
> newzeroplacesinmatrix):
> print("A módosított paraméterkészlet és az MM mátrix,
> az eddigiekböl."):

> cc:=MMcreate(k,allzeros):
> alist:=cc[1]:
> MM:=cc[2]:
> print("A Jacobi-azonosságból adódó egyenletek."):
> eqset:=setofeq(k,comm,MM):
> cc:=simpeqset(eqset):
> trivialzeros:=cc[1]:
> eqsetfinal:=cc[2]:
> print("Az egyenletek és a paraméterek száma."):
> numofeq:=nops(eqsetfinal):
> numofparameters:=nops(alist):
> print("A mátrix magterének dimenziója
> az egyenletek halmaza által."):
> kohom:=dimker(eqsetfinal,alist):
> print("Ez az a szám, mely a kommutátor kociklusok
> gyenge kohomológiáját jellemzi."):

k := 2

dimv := 6

numofparameters := 15

generators := [


0 −1 0 0

1 0 0 0

0 0 0 0

0 0 0 0

 ,


0 0 1 0

0 0 0 0

0 0 0 0

0 0 0 0

 ,


0 0 0 0

0 0 1 0

0 0 0 0

0 0 0 0

 ,


0 0 0 1

0 0 0 0

0 0 0 0

0 0 0 0

 ,


0 0 0 0

0 0 0 1

0 0 0 0

0 0 0 0

 ,


0 0 0 0

0 0 0 0

0 0 0 1

0 0 0 0

]

comm :=



0 3 −2 5 −4 0

−3 0 0 0 0 4

2 0 0 0 0 5

−5 0 0 0 0 0

4 0 0 0 0 0

0 −4 −5 0 0 0


”Az MM mátrix első megalkotása, és a paraméterek listája.”

alist := [a1, a2, a3, a4, a5, a6, a7, a8, a9, a10, a11, a12, a13, a14, a15]

MM :=



0 a1 a2 a3 a4 a5

−a1 0 a6 a7 a8 a9

−a2 −a6 0 a10 a11 a12

−a3 −a7 −a10 0 a13 a14

−a4 −a8 −a11 −a13 0 a15

−a5 −a9 −a12 −a14 −a15 0


„A null kohomológia a következő paramétereket adja 0-nak.”

cc := {1, 2, 3, 4} , {[1, 2], [1, 3], [1, 4], [1, 5]}
zeroplacesinmatrix := {[1, 2], [1, 3], [1, 4], [1, 5]}

„A módosított paraméterkészlet és az MM mátrix, az előzőekkel.”

alist := [a1, a2, a3, a4, a5, a6, a7, a8, a9, a10, a11]

MM :=



0 0 0 0 0 a1

0 0 a2 a3 a4 a5

0 −a2 0 a6 a7 a8

0 −a3 −a6 0 a9 a10

0 −a4 −a7 −a9 0 a11

−a1 −a5 −a8 −a10 −a11 0

„A Jacobi-azonosság a következő egyenleteket adja.”

eqset := {a8 = 0, a9 = 0, a11 = 0,−a5 = 0,−a9 = 0,−a10 = 0,−a4 + a6 = 0,

−a6 − a4 = 0, a6 + a4 = 0, a7 − a3 = 0}

„A null paraméterek és a null paraméterekkel egyszerűsített egyenletek.”

cc := {5, 8, 9, 10, 11} , {−a4 + a6 = 0, a6 + a4 = 0, a7 − a3 = 0}
zeroindex := {5, 8, 9, 10, 11}

newzeroplacesinmatrix := {[2, 6], [3, 6], [4, 5], [4, 6], [5, 6]}
allzeros := {[1, 2], [1, 3], [1, 4], [1, 5], [2, 6], [3, 6], [4, 5], [4, 6], [5, 6]}

„A módosított paraméterkészlet és az MM mátrix, az eddigiekből.”

alist := [a1, a2, a3, a4, a5, a6]

MM :=



0 0 0 0 0 a1

0 0 a2 a3 a4 0

0 −a2 0 a5 a6 0

0 −a3 −a5 0 0 0

0 −a4 −a6 0 0 0

−a1 0 0 0 0 0


„A Jacobi-azonosságból adódó egyenletek.”

eqset := {−a4 + a5 = 0,−a5 − a4 = 0, a5 + a4 = 0, a6 − a3 = 0}
trivialzeros := {}

eqsetfinal := {−a4 + a5 = 0, a5 + a4 = 0, a6 − a3 = 0}
„Az egyenletek és a paraméterek száma.”

numofeq := 3

numofparameters := 6

„A mátrix magterének dimenziója az egyenletek halmaza által.”

kohom := 3

„Ez az a szám, mely a kommutátor kociklusok gyenge kohomológiáját

jellemzi.”

Majd pedig lefuttattuk még a 4 + 1 tér-idő dimenziós esetben is, ám úgy
döntöttünk, hogy csak a legfontosabb adatokat közöljük, mivel ennek terje-
delme már szinte egy másik szakdolgozatot is kitenne. Így tehát a következők
adódtak a 4 tér térdimenziósra.

„A tér dimenziója és a kezdeti paraméterek száma:”, 4, 105

„A Jacobi-azonosság a következő paraméterek számát adja 0-nak.”, 84

„Az egyenletek és paraméterek száma:”, 10, 7

„A szabad paraméterek száma, melyek a kommutátor kociklusok gyenge

kohomológiáját jellemzik:”, 1

Végül pedig írtunk egy procedúrát, ami az előbbi procedúrák használa-
tával, csupán a tér dimenzió megadásával képes megadni nekünk az adott tér
dimenzióhoz tartozó Galilei-csoport szabad paramétereinek számát, azaz azt
a számot, mely a kommutátor kociklusok gyenge kohomológiáját jellemzi.

Továbbá érdekességképpen kiírattuk vele a következőket is rendre: a tér
dimenzióját, a kezdeti paraméterek számát, a gyenge kohomológia által kinul-
lázott paraméterek számát, a Jacobi-azonosság null paramétereinek számát,
az egyenletek és az új paraméterek számát, legvégül pedig a gyenge kohomo-
lógia jellemzésére használatos számot.

> cohom:=proc(kk)
> local k, dimv, numofinitialparameters, generators, comm,
> cc, alist, MM, zeroplacesinmatrix, firstzeros, eqset,
> zeroindex, numofparameters, newzeroplacesinmatrix,
> secondzeros, allzeros, trivialzeros, eqsetfinal, numofeq,
> kohom:
> k:=kk:
> dimv:=(1/2)*k^2+(3/2)*k+1:
> numofinitialparameters:=(1/2)*dimv*(dimv-1):
> generators:=create(k):
> comm:=commmatrix(k,generators):
> cc:=Mcreate(k):
> alist:=cc[1]:
> MM:=cc[2]:
> cc:=zerocohom(k,comm,MM,alist):

> zeroplacesinmatrix:=cc[2]:
> firstzeros:=nops(zeroplacesinmatrix):
> cc:=MMcreate(k,zeroplacesinmatrix):
> alist:=cc[1]:
> MM:=cc[2]:
> eqset:=setofeq(k,comm,MM):
> cc:=simpeqset(eqset):
> zeroindex:=cc[1]:
> secondzeros:=nops(zeroindex):
> newzeroplacesinmatrix:=zerosinM(k,MM,zeroindex):
> allzeros:=‘union‘(zeroplacesinmatrix,
> newzeroplacesinmatrix):
> cc:=MMcreate(k,allzeros):
> alist:=cc[1]:
> MM:=cc[2]:
> eqset:=setofeq(k,comm,MM):
> cc:=simpeqset(eqset):
> trivialzeros:=cc[1]:
> eqsetfinal:=cc[2]:
> numofeq:=nops(eqsetfinal):
> numofparameters:=nops(alist):
> kohom:=dimker(eqsetfinal,alist):
> return k, numofinitialparameters, firstzeros,
> secondzeros, numofeq, numofparameters, kohom:
> end proc:

Legvégül pedig egy for ciklus használatával lefuttattuk és kiírattuk az
előbb részletezett cohom procedúrát 1-től 10-ig.

> for e to 10 do print(cohom(e))
> end do:

1, 3, 1, 0, 0, 2, 2

2, 15, 4, 5, 3, 6, 3

3, 45, 9, 30, 6, 6, 1

4, 105, 14, 84, 10, 7, 1

5, 210, 20, 185, 10, 5, 1

6, 378, 27, 345, 15, 6, 1

7, 630, 35, 588, 21, 7, 1

8, 990, 44, 938, 28, 8, 1

9, 1485, 54, 1422, 36, 9, 1

10, 2145, 65, 2070, 45, 10, 1

Ezekből pedig az látszik, hogy a tér dimenzió növelésével nagyon gyorsan
nő a kezdeti paraméterek száma, melyekből a legtöbb a Jacobi-azonosság null
paramétereinek meghatározásakor esik ki. Valamint megfigyelhető, hogy a
legalább 3 tér dimenziós Galilei-csoportok esetében a kommutátor kociklusok
gyenge kohomológiái már csupán egyetlen paramétertől függnek.

4. Összefoglaló

A két és három térdimenziós Galilei-csoport gyenge kohomológiájának
jellemzése már ismert volt. Jelen szakdolgozattal az volt a célunk, hogy
magasabb térdimenziós esetekben is megadjuk a Galilei-csoport gyenge ko-
homológiáinak jellemzésére alkalmas paramétereket.

Az 1. fejezetben bevezettük a Galilei-csoport definícióját, és megismer-
kedtünk néhány alaptulajdonságával [1] és [5] könyvek, valamint a [6] szak-
dolgozat segítségével. Definiáltuk a témakör tárgyalásához szükséges alapfo-
galmakat a [3] alapján, valamint meghatároztuk a Lie-algebrát a [4] szócikk
felhasználásával. Továbbá a Galilei-csoport megértéséhez szükséges fogal-
makat vezettünk be [5] és [6] alapján. Megismerkedtünk a félcsoportokkal,
az egyparaméteres részcsoportokkal, a kommutátor kociklusokkal. Ezeket
a fogalmakat egy egyszerű példán, a speciális lineáris csoporton keresztül
szemléltettük a [2] könyv alapján.

A 2. fejezetben a legfeljebb három térdimenziós Galilei-csoport kommu-
tátor kociklusainak gyenge kohomológiáit számítottuk végig. A 3. fejezetben
pedig egy számítógépes programot prezentáltunk, amit arra írtunk, hogy ma-
gasabb tér dimenziós esetekben is megállapíthassuk a Galilei-csoport gyenge
kohomológiáinak jellemzésére alkalmas paramétereket, melyeket a program
segítségével 10 tér dimenzióig kiszámítottunk. Láttuk, hogy a legalább 3
térdimenziós esetekben a függő paraméterek száma 1 volt.

